
import "util.set"; 
import "util.type"; 
import "util.dictionary"; 

import "strategies.property_importance"; 

signature PROPERTY = 
sig 
    type property; 

    exception ParseError; 

    val compare : property * property -> order; 
    val propertyMatch : property * property -> bool; 
    val nameOf : property -> string; 
    val toString : property -> string; 
    val fromString : string -> property; 
end; 

structure Property :> PROPERTY = 
struct 

datatype property = Simple of string 
                  | Typed of (string * Type.T) 
                  | Attr of (string * string list); 

exception ParseError; 

fun nameOf (Simple s) = s 
  | nameOf (Typed (s,_)) = s 
  | nameOf (Attr (s,_)) = s; 

(*A lexicographic order for the property type. The name takes precedence, then 
  the KIND (Simple, Typed, Attr), and in the end the type or attribute list. 
  Useful for putting it into a dictionary; not for much else*) 
fun compare (Simple s, Simple s') = String.compare (s, s') 
  | compare (Typed (s,t), Typed (s',t')) = let val c = String.compare (s, s') 
                                           in if c = EQUAL 
                                              then Type.compare (t,t') 
                                              else c 
                                           end 
  | compare (Attr (s,a), Attr (s',a')) = let val c = String.compare (s, s') 
                                           in if c = EQUAL 
                                              then List.collate String.compare (a,a') 
                                              else c 
                                           end 
  | compare (p,p') = let val c = String.compare (nameOf p, nameOf p') 
                     in if c = EQUAL 
                        then case (p,p') of (Simple _, _) => LESS 
                                          | (_, Simple _) => GREATER 
                                          | (Typed _, _) => LESS 
                                          | (_, Typed _) => GREATER 
                                          | _ => raise Match 
                        else c 
                     end  ; 

(*propertyMatch is meant to be used for finding whether a correspondence holds 
  without the need to have type or attribute information, and type-matching when 
  there is type information *) 
fun propertyMatch (Simple s, p) = (s = nameOf p) 
  | propertyMatch (p, Simple s) = (s = nameOf p) 
  | propertyMatch (Typed (s,t), Typed (s',t')) = (s = s' andalso Type.match t t') 
  | propertyMatch (Attr (s,_), Attr (s',_)) = (s = s') 
  | propertyMatch _ = false 

fun toString (Simple s) = s 
  | toString (Typed (s,t)) = s ^ " : " ^ (Type.toString t) 
  | toString (Attr (s,a)) = s ^ " : {" ^ (String.concat (intersperse ", " a)) ^ "}"; 

(*as-is, fromString is an ugly function. Very ad-hoc.*) 
fun fromString x = 
  if String.isPrefix "pattern-" x then Simple x else 
      case map stringTrim (String.tokens (fn c => c = #":") x) of 
         [r,s] => 
            if String.substring (s,0,1) = "{" then 
                if s = "{}" then Attr (r,[]) 
                else let 
                          fun dropEnds [] = [] 
                            | dropEnds [x] = [] 
                            | dropEnds [x, y] = [] 
                            | dropEnds (x::xs) = List.rev (List.tl (List.rev xs)); 
                          val s' = String.implode (dropEnds (String.explode s)); 
                      in 
                        Attr (r,map stringTrim (String.tokens (fn c => c = #";") s')) 
                      end 
            else Typed (r, Type.fromString s) 
       | [r] => Simple r 
       | _ => raise Match; 

end; 

signature QPROPERTY = 
sig 
    type property; 

    exception ParseError; 

    val compare : property * property -> order; 
    val toString : property -> string; 
    val fromString : string -> property; 
    val toPair : property -> (Property.property * Importance.importance); 
    val fromPair : (Property.property * Importance.importance) -> property; 
    val withoutImportance : property -> Property.property; 
end; 

structure QProperty :> QPROPERTY = 
struct 

type property = (Property.property * Importance.importance); 
exception ParseError; 

val compare = cmpJoin Property.compare Importance.compare; 
fun toString (p, i) = "(" ^ (Property.toString p) ^ ", " ^ (Importance.toString i) ^ ")"; 
fun fromString s = 
    let 
        val splitter = fn c => c = #"(" orelse  c = #")" orelse c = #","; 
    in 
        case String.tokens splitter s of 
            [a, b] => (case Importance.fromString b of 
                           SOME b' => (Property.fromString a, b') 
                         | NONE => raise ParseError) 
          | _ => raise ParseError 
    end; 
fun toPair (s, i) = (s, i); 
fun fromPair (s, i) = (s, i); 
fun withoutImportance (s, _) = s; 

end; 

structure PropertySet = Set(struct 
                             type t = Property.property; 
                             val compare = Property.compare; 
                             val fmt = Property.toString; 
                             end); 
structure PropertyDictionary = Dictionary(struct 
                                           type k = Property.property; 
                                           val compare = Property.compare; 
                                           val fmt = Property.toString; 
                                           end); 
structure QPropertySet = Set(struct 
                              type t = QProperty.property; 
                              val compare = QProperty.compare; 
                              val fmt = QProperty.toString; 
                              end); 
import "util.set"; 
import "util.dictionary"; 

import "strategies.property"; 
import "strategies.property_importance"; 

signature CORRESPONDENCE = 
sig 

    structure S: SET; 
    structure D: DICTIONARY; 

    exception ParseError 

    datatype 'a corrformula = Atom of 'a 
                            | Neg of 'a corrformula 
                            | Conj of 'a corrformula * 'a corrformula 
                            | Disj of 'a corrformula * 'a corrformula; 
    type correspondence = Property.property corrformula * Property.property corrformula * 
real; 

    val equal : correspondence -> correspondence -> bool; 
    val match : S.t S.set -> S.t S.set -> correspondence -> bool; 
    val matchingIntersectionLeft : S.t S.set -> S.t S.set -> S.t S.set; 
    val sameProperties : correspondence -> correspondence -> bool; 
    val matchingProperties : correspondence -> correspondence -> bool; 
    val strength : correspondence -> real; 

    val toString : correspondence -> string; 
    val fromString : (string -> Property.property) -> string -> correspondence; 
end; 

structure Correspondence : CORRESPONDENCE = 
struct 

structure S = PropertySet; 

structure D = PropertyDictionary; 

exception ParseError; 

datatype 'a corrformula = Atom of 'a 
                        | Neg of 'a corrformula 
                        | Conj of 'a corrformula * 'a corrformula 
                        | Disj of 'a corrformula * 'a corrformula; 
type correspondence = Property.property corrformula * Property.property corrformula * real; 

fun formulamap f (Atom a) = Atom (f a) 
  | formulamap f (Neg a) = Neg (formulamap f a) 
  | formulamap f (Conj (a, b)) = Conj (formulamap f a, formulamap f b) 
  | formulamap f (Disj (a, b)) = Disj (formulamap f a, formulamap f b); 

fun emptyIntn a b = S.isEmpty (S.intersection a b); 

fun strength (_, _, s) = s; 

fun filterMatches p ps = S.filter (fn v => Property.propertyMatch (p,v)) ps; 

fun propertyMatches p ps = not (S.isEmpty (filterMatches p ps)); 

fun allPropertiesMatch ps ps' = 
    let val contained = S.map (fn p => propertyMatches p ps') ps; 
    in all contained end; 

fun somePropertiesMatch ps ps' = 
    let val contained = S.map (fn p => propertyMatches p ps') ps; 
    in any contained end; 

fun joinSets [] = S.empty () 
  | joinSets (h::t) = S.union h (joinSets t) 

(* finds matches between two property sets, 
but only returns the matches of the left (ps) *) 
fun matchingIntersectionLeft ps ps' = 
    let val x = S.map (fn p => filterMatches p ps) ps' 
    in foldr (fn (a,b) => S.union a b) (S.empty ()) x 
    end; 

fun match qs rs ((qp, qn), (rp, rn), _) = 
    (allPropertiesMatch qp qs) andalso 
    (allPropertiesMatch rp rs) andalso 
    (not (somePropertiesMatch qn qs)) andalso 
    (not (somePropertiesMatch rn rs)); 

fun biMatch (ps,ps') = allPropertiesMatch ps ps' andalso allPropertiesMatch ps' ps 

fun matchingProperties ((qp, qn), (rp, rn), _) ((qp', qn'), (rp', rn'), _) = 
    biMatch (qp, qp') andalso 
    biMatch (qn, qn') andalso 
    biMatch (rp, rp') andalso 
    biMatch (rn, rn'); 

fun sameProperties ((qp, qn), (rp, rn), _) ((qp', qn'), (rp', rn'), _) = 
    S.equal (qp, qp') andalso 
    S.equal (qn, qn') andalso 
    S.equal (rp, rp') andalso 
    S.equal (rn, rn'); 

fun equal c c' = sameProperties c c' andalso Real.== ((strength c), (strength c')); 

fun toString ((qp, qn), (rp, rn), s) = 
    "((" ^ 
    (S.toString qp) ^ 
    ", " ^ 
    (S.toString qn) ^ 
    "), (" ^ 
    (S.toString rp) ^ 
    ", " ^ 
    (S.toString rn) ^ 
    ")) -> " ^ 
    (Real.toString s); 

fun normalise (Atom a) = Atom a 
  | normalise (Neg a) = 
    let 
        val a' = normalise a; 
    in 
        case a' of 
            Atom t => Neg (Atom t) 
          | Neg t => t 
          | Conj (t, u) => normalise (Disj (Neg t, Neg u)) 
          | Disj (t, u) => normalise (Conj (Neg t, Neg u)) 
    end 
  | normalise (Conj (a, b)) = 
    let 
        val a' = normalise a; 
        val b' = normalise b; 
    in 
        case (a', b') of 
            (Disj (t, u), v) => normalise (Disj (Conj (t, v), 
                                                 Conj (u, v))) 
          | (t, Disj(u, v)) => normalise (Disj (Conj (t, u), 
                                                Disj (Conj (t, v)))) 
          | (u, v) => Conj (u, v) 
    end 
  | normalise (Disj (a, b)) = 
    let 
        val a' = normalise a; 
        val b' = normalise b; 
    in 
        Disj (a', b') 
    end; 

fun fromString propMaker s = 
    let 
        val temp = (Atom (propMaker s), Atom (propMaker s), 1.0); 
        val parts = String.tokens (fn c => c = #",") s; 
        val [leftString, rightString, valString] = parts 
                                                   handle Match => raise Match; 

        fun tokenize string = 
            let 
                fun cluster [] xs = cluster [[]] xs 
                  | cluster cs [] = cs 
                  | cluster (c::cs) (x::xs) = 
                    case x of 
                        #"(" => cluster ([]::[#"("]::c::cs) xs 
                      | #")" => cluster ([]::[#")"]::c::cs) xs 
                      | s => if Char.isSpace s 
                             then cluster ([]::c::cs) xs 
                             else cluster ((s::c)::cs) xs; 
            in 
                List.rev (map (String.implode o List.rev) 
                              (List.filter 
                                   (fn cs => not (List.null xs)) 
                                   (cluster [[]] (String.explode string)))) 
            end; 

        fun nextToken [] = NONE 
          | nextToken (x::xs) = SOME x; 
        fun remainingTokens [] = [] 
          | remainingTokens (x::xs) = xs; 

        fun expect s [] = raise ParseError 
          | expect s (x::xs) = if String.toLower(s) = String.toLower(x) 
                               then (s, xs) 
                               else raise ParseError; 

        fun parseAtom [] = raise ParseError 
          | parseAtom (x::xs) = if x = "(" 
                                    then raise ParseError 
                                else if x = ")" 
                                    then raise ParseError 
                                else if String.toLower(x) = "and" 
                                    then raise ParseError 
                                else if String.toLower(x) = "or" 
                                    then raise ParseError 
                                else if String.toLower(x) = "not" 
                                    then raise ParseError 
                                else (x, xs) 
        and parseNeg [] = raise ParseError 
          | parseNeg xs = 
            let 
                val (neg, resta) = expect "not" xs; 
                val (atom, restb) = parseCForm resta; 
            in 
                (Neg atom, restb) 
            end 
        and parseConj [] = raise ParseError 
          | parseConj xs = 
            let 
                val (left, resta) = parseBForm xs; 
                val (andtok, restb) = expect "and" resta; 
                val (right, restc) = parseAForm restb; 
            in 
                (Conj (left, right), restc) 
            end 
        and parseDisj [] = raise ParseError 
          | parseDisj xs = 
            let 
                val (left, resta) = parseAForm xs; 
                val (ortok, restb) = expect "or" resta; 
                val (right, restc) = parseFormula restb; 
            in 
                (Disj (left, right), restc) 
            end 
        and parseFormula xs = parseDisj xs 
                              handle ParseError => parseAForm xs 
        and parseAForm xs = parseConj xs 
                            handle ParseError => parseBForm xs 
        and parseBForm xs = parseNeg xs 
                            handle ParseError => parseCForm xs 
        and parseCForm xs = parseAtom xs 
                            handle ParseError => 
                                   let 
                                       val (lparn, resta) = expect "(" xs; 
                                       val (toks, restb) = toCloseParen resta; 
                                   in 
                                       case restb of 
                                           [] => parseForm toks 
                                         | _ => raise ParseError 
                                   end; 

        fun parse tokens = 
            let 
                val (tree, tokens) = parseFormula tokens; 
            in 

                if (List.null tokens) then tree 
                else raise Match 
            end; 

        val read = (normalise o parse o tokenize); 
    in 
        (read leftString, read rightString, Real.fromString valString) 
    end; 

end; 
signature IMPORTANCE = 
sig 

    datatype importance = Noise | Zero | Low | Medium | High; 

    val compare : (importance * importance) -> order; 
    val fromString : string -> importance option; 
    val toString : importance -> string; 

end; 

structure Importance : IMPORTANCE = 
struct 

datatype importance = Noise | Zero | Low | Medium | High; 

fun compare (a, b) = let 
    fun ordify Noise = ~1 
      | ordify Zero = 0 
      | ordify Low = 1 
      | ordify Medium = 2 
      | ordify High = 3; 
in 
    Int.compare (ordify a, ordify b) 
end; 

fun fromString "Noise" = SOME Noise 
  | fromString "Zero" = SOME Zero 
  | fromString "Low" = SOME Low 
  | fromString "Medium" = SOME Medium 
  | fromString "High" = SOME High 
  | fromString _ = NONE; 

fun toString Noise = "Noise" 
  | toString Zero = "Zero" 
  | toString Low = "Low" 
  | toString Medium = "Medium" 
  | toString High = "High"; 

end; 
(* 
The following functions handle how properties are generated from the "property" 
table. For example, they key "operators" then lists a collection of operators. 
To read this, we use the "readCollection" function and will prepend each of the 
operators with the string "op-". Compare this with basic labels, which simply 
return the one thing that is there, in a list, ready to prepend to. Bools are 
simplest, either returning an empty list (false), or a list containing the empty 
string (true) to generate either the key, or nothing, as a property. 
A concrete example: From the table 
    operators        +, -, *, \sqrt 
    sentential       true 
    logic-power      2 
we would generate the properties 
    op-+, op--, op-*, op-\sqrt, sentential, logic-power-2 
*) 

import "strategies.property_importance"; 
import "strategies.property_tables"; 

structure PropertyReader = 
struct 

fun bool str = if ((String.implode (map Char.toLower (String.explode str))) = "true") 
               then [""] else []; 

fun label str = [str]; 

fun collection str = if str = "NONE" then [] 
                     else map stringTrim (String.tokens (fn c => c = #",") str); 

fun dimension str = 
    let 
        fun parseDimProps s = if s = "{}" then [] 
                              else let 
                                  fun dropEnds [] = [] 
                                    | dropEnds [x] = [] 
                                    | dropEnds [x, y] = [] 
                                    | dropEnds (x::xs) = List.rev (List.tl (List.rev xs)); 
                                  val s' = String.implode (dropEnds (String.explode s)); 
                              in 
                                  map stringTrim (String.tokens (fn c => c = #";") s') 
                              end; 
        fun createPairs dimval = 
            let 
                val parts = map stringTrim (String.tokens (fn c => c = #":") dimval); 
            in 
                case parts of 
                    [x, y] => (x, parseDimProps y) 
                  | _ => raise PropertyTables.TableError 
                               ("Unable to read dimensions from " ^ dimval) 
            end; 
        val dimensions = collection str; 
        val dimensionsWithValues = map createPairs dimensions; 
        val dimensionsSplitOut = map (fn (x, y) => 
                                         map (fn z => x ^ "-" ^ z) y) 
                                     dimensionsWithValues; 
        val dimensionsNoLabels = map (fn (x, y) => 
                                         map (fn z:string => z) y) 
                                     dimensionsWithValues; 
    in 
        List.foldr (fn (a, b) => a @ b) [] (dimensionsSplitOut @ dimensionsNoLabels) 
    end; 
     
end; 

let 
    open Importance; 
    fun stripImportance vals = map (fn (l, (f, p, i)) => (l, (f, p))) vals; 

    val RSProperties = [ 
        ("mode", 
         (PropertyReader.collection, "mode-")), 
        ("grammar-imports", 
         (PropertyReader.collection, "import-")), 
        ("grammatical-complexity", 
         (PropertyReader.label,  "grammatical-complexity-")), 
        ("rigorous", 
         (PropertyReader.bool, "rigorous")), 
        ("knowledge-manipulation-system", 
         (PropertyReader.bool, "knowledge-manipulation-system")), 
        ("facts", 
         (PropertyReader.collection, "fact-")), 
        ("fact-imports", 
         (PropertyReader.collection, "import-")), 
        ("tactics", 
         (PropertyReader.collection, "tactic-")), 
        ("logic-power", 
         (PropertyReader.label, "logic-power-")), 
        ("standard-accessibility-manipulations", 
         (PropertyReader.collection, "accessible-manipulation-")), 
        ("accessible-facts", 
         (PropertyReader.bool, "accessible-facts")), 
        ("accessible-tactics", 
         (PropertyReader.bool, "accessible-tactics")), 
        ("accessible-grammatical-constructors", 
         (PropertyReader.bool, "accessible-grammatical-constructors")), 
        ("editable-external-memory", 
         (PropertyReader.bool, "editable-external-memory")), 
        ("physical-dimension-use", 
         (PropertyReader.dimension, "dimension-use-")), 
        ("grammatical-dimensionality", 
         (PropertyReader.label, "grammatical-dimensionality-")), 
        ("grammatical-granularity", 
         (PropertyReader.label, "grammatical-granularity-")), 
        ("mean-branching-factor", 
         (PropertyReader.label, "mean-branching-factor-")), 
        ("mean-solution-depth", 
         (PropertyReader.label, "mean-solution-depth-")), 
        ("pr-distinct-state-change", 
         (PropertyReader.label, "pr-distinct-state-change-")), 
        ("pr-valid-state-change", 
         (PropertyReader.label, "pr-valid-state-change-")), 
       ("types", 
         (PropertyReader.collection, "type-")), 
       ("tokens", 
         (PropertyReader.collection, "token-")), 
       ("operators", 
         (PropertyReader.collection, "token-")), 
       ("relations", 
         (PropertyReader.collection, "token-")), 
       ("patterns", 
         (PropertyReader.collection, "pattern-")) 
    ]; 
    val QProperties = [ 
        ("error-allowed", 
         (PropertyReader.label, "error-allowed-", High)),  (*previously rigorous*) 
        ("answer-type", 
         (PropertyReader.collection, "type-", High)), (*previously question-value-type*) 
        ("instrumental-tokens", 
         (PropertyReader.collection, "token-", Medium)), 
        ("instrumental-types", 
         (PropertyReader.collection, "type-", Medium)), 
        ("instrumental-patterns", 
         (PropertyReader.collection, "pattern-", Medium)), 
        ("instrumental-facts", 
         (PropertyReader.collection, "fact-", Medium)), 
        ("instrumental-tactics", 
         (PropertyReader.collection, "tactic-", Medium)), 
        ("relevant-tokens", 
         (PropertyReader.collection, "token-", Low)), 

        ("relevant-related-tokens", 
         (PropertyReader.collection, "token-", Low)), 
        ("num-statements", 
         (PropertyReader.label, "num-statements-", Zero)), 
        ("num-tokens", 
         (PropertyReader.label, "num-tokens-", Zero)), 
        ("num-distinct-tokens", 
         (PropertyReader.label, "num-distinct-tokens-", Zero)), 
        ("noise-tokens", 
         (PropertyReader.collection, "token-", Noise)), 
        ("noise-related-tokens", 
         (PropertyReader.collection, "token-", Noise)) 
    ]; 
    val QandRSProperties = [ 
    ]; 
in 
    PropertyTables.setQGenerators 
        (QProperties @ QandRSProperties); 
    PropertyTables.setRSGenerators 
        (RSProperties @ (stripImportance QandRSProperties)) 
end; 
import "util.logging"; 
import "util.set"; 
import "util.dictionary"; 
import "util.csv"; 

import "strategies.property"; 
import "strategies.property_importance"; 
import "strategies.property_correspondence"; 

signature PROPERTYTABLES = 
sig 
    exception TableError of string; 

    structure FileDict : DICTIONARY; 

    type qgenerator = (string -> string list) * string * Importance.importance; 
    type rsgenerator = (string -> string list) * string; 

    val loadCorrespondenceTable : string -> Correspondence.correspondence list; 
    val loadQuestionTable : string -> (FileDict.k, QPropertySet.t QPropertySet.set) 
FileDict.dict; 
    val loadRepresentationTable : string -> (FileDict.k, PropertySet.t PropertySet.set) 
FileDict.dict; 

    val setQGenerator : (string * qgenerator) -> unit; 
    val setQGenerators : (string * qgenerator) list -> unit; 
    val setRSGenerator : (string * rsgenerator) -> unit; 
    val setRSGenerators : (string * rsgenerator) list -> unit; 

end; 

structure PropertyTables : PROPERTYTABLES = 
struct 

exception TableError of string; 

structure SQ = QPropertySet; 
structure S = PropertySet; 
val qset' = SQ.fromList; 
val set' = S.fromList; 

structure D = PropertyDictionary; 
val dict' = D.fromPairList; 
fun getValue d k = SOME (D.get d k) 
                   handle D.KeyError => NONE; 

structure FileDict = Dictionary(struct 
                                 type k = string; 
                                 val compare = String.compare; 
                                 val fmt = (fn s => s); 
                                 end); 
val filedict' = FileDict.fromPairList; 

structure GenDict = Dictionary(struct 
                                 type k = string; 
                                 val compare = String.compare; 
                                 val fmt = (fn s => s); 
                                 end); 
val gdict' = GenDict.fromPairList; 

structure CSVLiberal = CSVIO(struct val delimiters = [#","]; 
                                    val newlines = ["\r", "\n", "\r\n"]; 
                             end); 

type qgenerator = (string -> string list) * string * Importance.importance; 
type rsgenerator = (string -> string list) * string; 

datatype CorrTree = Prop of string 
                  | Neg of CorrTree 
                  | Conj of CorrTree * CorrTree 
                  | Disj of CorrTree * CorrTree; 

fun readCorrespondence qpString rspString strengthString = 
    let 
        fun tokenize string = 
            let 
                fun cluster [] xs = cluster [[]] xs 
                  | cluster cs [] = cs 
                  | cluster (c::cs) (x::xs) = 
                    case x of 
                        #"(" => cluster ([]::[#"("]::c::cs) xs 
                      | #")" => cluster ([]::[#")"]::c::cs) xs 
                      | s => if Char.isSpace s 
                             then cluster ([]::c::cs) xs 
                             else cluster ((s::c)::cs) xs; 
            in 
                List.rev 
                    (map (String.implode o List.rev) 
                         (List.filter 
                              (fn cs => not (List.null cs)) 
                              (cluster [[]] (String.explode string)))) 
            end; 

        fun parseAtom tokens = 
            let 
                val nextTok = if (List.null tokens) then NONE 
                              else (SOME (List.hd tokens)); 
            in 
                case nextTok of 
                    SOME "(" => 
                    let 
                        val useTokens = List.tl tokens; 
                        val (disjtree, remainingTokens) = parseDisj useTokens; 
                        val closePar = if (List.null remainingTokens) then NONE 
                                       else (SOME (List.hd remainingTokens)); 
                    in 
                        case closePar of 
                            SOME ")" => (disjtree, (List.tl remainingTokens)) 
                          | SOME other => raise TableError "Missing closing paren" 
                          | NONE => (disjtree, []) 
                    end 
                  | SOME other => (Prop other, (List.tl tokens)) 
                  | NONE => raise TableError "Unexpected EOF" 
            end 
        and parseNeg tokens = 
            let 
                val nextTok = if (List.null tokens) then NONE 
                              else (SOME (List.hd tokens)); 
            in 
                case nextTok of 
                    SOME "NOT" => 
                    let 
                        val useTokens = List.tl tokens; 
                        val (atomTree, remainingTokens) = parseAtom useTokens; 
                    in 
                        (Neg atomTree, remainingTokens) 
                    end 
                  | SOME other => parseAtom tokens 
                  | NONE => raise TableError "Unexpected EOF" 
            end 
        and parseConj tokens = 
            let 
                val (readIn, nextTokens) = parseNeg tokens; 
                val nextTok = if (List.null nextTokens) then NONE 
                              else (SOME (List.hd nextTokens)); 
            in 
                case nextTok of 
                    SOME "AND" => 
                    let 
                        val useTokens = List.tl nextTokens; 
                        val (conjtree, remainingTokens) = parseConj useTokens; 
                    in 
                        (Conj (readIn, conjtree), remainingTokens) 
                    end 
                  | SOME other => (readIn, nextTokens) 
                  | NONE => (readIn, []) 
            end 
        and parseDisj tokens = 
            let 
                val (readIn, nextTokens) = parseConj tokens; 
                val nextTok = if (List.null nextTokens) then NONE 
                              else (SOME (List.hd nextTokens)); 
            in 
                case nextTok of 
                    SOME "OR" => 
                    let 
                        val useTokens = List.tl nextTokens; 
                        val (distree, remainingTokens) = parseDisj useTokens; 
                    in 
                        (Disj (readIn, distree), remainingTokens) 
                    end 
                  | SOME other => (readIn, nextTokens) 
                  | NONE => (readIn, []) 
            end; 

        fun parse tokens = 
            let 
                val (tree, tokens) = parseDisj tokens; 
            in 

                if (List.null tokens) then tree else raise TableError "Unprocessed content" 
            end; 

        fun normalise (Prop s) = Prop s 
          | normalise (Neg s) = 
            let 
                val s' = normalise s; 
            in 
                case s' of 
                    Prop k => Neg (Prop k) 
                  | Neg k => normalise k 
                  | Conj (a, b) => normalise (Disj (Neg a, Neg b)) 
                  | Disj (a, b) => normalise (Conj (Neg a, Neg b)) 
            end 
          | normalise (Conj (a, b)) = 
            let 
                val a' = normalise a; 
                val b' = normalise b; 
            in 
                case (a', b') of 
                    (Disj (u, v), w) => normalise (Disj (Conj (u, w), 
                                                         Conj (v, w))) 
                  | (u, Disj (v, w)) => normalise (Disj (Conj (u, v), 
                                                         Conj (u, w))) 
                  | (u, v) => Conj (u, v) 
            end 
          | normalise (Disj (a, b)) = 
            let 
                val a' = normalise a; 
                val b' = normalise b; 
            in 
                Disj (a', b') 
            end; 

        fun setify (Prop s) = (set' [Property.fromString s], S.empty ()) 
          | setify (Neg (Prop s)) = (S.empty (), set' [Property.fromString s]) 
          | setify (Neg _) = raise TableError 
                                   "Correspondences incorrectly normalised" 
          | setify (Conj (a, b)) = 
            let 
                fun ConjFlatten (Prop s) = [Prop s] 
                  | ConjFlatten (Neg a) = [Neg a] 
                  | ConjFlatten (Conj (a, b)) = (ConjFlatten a) @ (ConjFlatten b) 
                  | ConjFlatten (Disj _) = raise TableError 
                                                 "Correspondences incorrectly normalised"; 
                fun isPos (Prop s) = true 
                  | isPos _ = false; 
                fun isNeg (Neg a) = true 
                  | isNeg _ = false; 
                fun stripTreeness (Prop s) = Property.fromString s 
                  | stripTreeness (Neg (Prop s)) = Property.fromString s 
                  | stripTreeness _ = raise TableError 
                                            "Correspondences incorrectly normalised"; 
                val flattened = ConjFlatten (Conj (a, b)); 
                val positives = set' (map stripTreeness (List.filter isPos flattened)); 
                val negatives = set' (map stripTreeness (List.filter isNeg flattened)); 
            in 
                (positives, negatives) 
            end 
          | setify (Disj _) = raise TableError 
                                              "Correspondences incorrectly normalised"; 

        val read = (normalise o parse o tokenize); 

        fun strength str = 
            case (Real.fromString str) of 
                SOME v => v 
              | NONE => raise TableError "Correspondence strength is not a float"; 

        fun genCorr (Disj (a, b)) t2 v = 
            let 
                val expa = genCorr a t2 v; 
                val expb = genCorr b t2 v; 
                val expc = genCorr (normalise (Conj (a, b))) t2 (~v); 
            in 
                expa @ expb @ expc 
            end 
          | genCorr t1 (Disj (a, b)) v = 
            let 
                val expa = genCorr t1 a v; 
                val expb = genCorr t1 b v; 
                val expc = genCorr t1 (normalise (Conj (a, b))) (~v); 
            in 
                expa @ expb @ expc 
            end 
          | genCorr t1 t2 v = [(t1, t2, v)]; 

        val qpTree = read qpString; 
        val rspTree = read rspString; 
        val corVal = strength strengthString; 
    in 
        map 
            (fn (t1, t2, v) => (setify t1, setify t2, v)) 
            (genCorr qpTree rspTree corVal) 
    end; 

fun loadCorrespondenceTable filename = 
    let 
        fun makeRow [x, y] = (readCorrespondence x y "1.0") 
          | makeRow [x, y, z] = (readCorrespondence x y z) 
          | makeRow r = raise TableError 
                              ("Correspondence table entry malformed: " ^ 
                               (listToString (fn s => s) r)); 
        val csvFile = CSVLiberal.openIn filename; 
        val csvData = CSVLiberal.input csvFile; 
    in 
        List.foldr (fn (r, xs) => (makeRow r) @ xs) [] csvData 
    end 
    handle IO.Io e => (Logging.error ("ERROR: File '" ^ filename ^ "' could not be 
loaded\n"); 
                       raise (IO.Io e)) 
         | TableError reason => ( 
             Logging.error ("ERROR: CSV parsing failed in file '" ^ filename ^ "'\n"); 
             Logging.error ("       " ^ reason ^ "\n"); 
             raise TableError reason 
         ); 

(* 
We provide a way to extend the known set of property generators with custom 
generation functions. These take the form of 
    (property-key, (generator, prefix, importance)) 
where property-key and prefix are strings, importance is an importance, and 
generator is a function. To add them, use either setGenerators with a list, 
or setGenerator with a tuple. Avoid running map over setGenerator, as it is 
faster to use the predefined 'plural'. 
Note that there are two types: QGenerators and RSGenerators. The first include 
the importance part, while the second do not. This is because there is no 
concept of importance when dealing with the RS in abstract terms. The question 
generators give the default importance of a property, but the property tables 
can over-ride this importance by specifying it in a third column. 
*) 
val qPropertyKeyMap = ref (GenDict.empty ()); 
fun setQGenerators new = 
    let 
        val _ = qPropertyKeyMap := GenDict.union (gdict' new) (!qPropertyKeyMap); 
    in () end; 
fun setQGenerator new = 
    let 
        val _ =  GenDict.insert (!qPropertyKeyMap) new; 
    in () end; 

val rPropertyKeyMap = ref (GenDict.empty ()); 
fun setRSGenerators new = 
    let 
        val _ = rPropertyKeyMap := GenDict.union (gdict' new) (!rPropertyKeyMap); 
    in () end; 
fun setRSGenerator new = 
    let 
        val _ = GenDict.insert (!rPropertyKeyMap) new; 
    in () end; 

fun loadQorRSPropertiesFromFile sets parseRow genProps filename  = 
    let 
        val (setEmpty, setUnion) = sets; 
        val csvFile = CSVLiberal.openIn filename; 
        val csvDataWithHeader = CSVLiberal.input csvFile; 
        val csvHeader = 
            (case (List.hd csvDataWithHeader) of 
                 [] => raise TableError "Table header blank" 
               | [h] => h 
               | [h, ""] => h 
               | _ => raise TableError "Missing table header") 
            handle List.Empty => raise TableError "Table is empty"; 
        val csvData = (List.tl csvDataWithHeader) 
                      handle List.Empty => raise TableError "Table is empty"; 

        val properties = List.foldr 
                             (fn (r, xs) => setUnion (genProps r) xs) 
                             (setEmpty ()) 
                             (map parseRow csvData); 
    in 
        [(csvHeader, properties)] 
    end 
    handle IO.Io e => (Logging.error ("ERROR: File '" ^ filename ^ "' could not be 
loaded\n"); 
                       raise (IO.Io e)) 
         | TableError reason => ( 
             Logging.error ("ERROR: CSV parsing failed in file '" ^ filename ^ "'\n"); 
             Logging.error ("       " ^ reason ^ "\n"); 
             raise TableError reason 
         ); 

fun loadQuestionTable filename = let 
    val sets = (SQ.empty, SQ.union); 
    fun parseImportance s = case Importance.fromString s of 
                                SOME i => i 
                              | NONE => raise TableError ("Unknown importance '" ^ s ^ "'"); 
    fun parseRow [x, y] = (x, y, NONE) 
      | parseRow [x, y, z] = (x, y, SOME (parseImportance z)) 
      | parseRow _ = raise TableError "Malformed question property entry"; 

    fun genProps (key, args, overrideImportance) = 
        let 
            fun findQGenerator k = SOME (GenDict.get (!qPropertyKeyMap) k) 
                                   handle GenDict.KeyError => NONE; 
            val (valparser, keypre, defaultImportance) = 
                case (findQGenerator key) of 
                    SOME kt => kt 
                  | NONE => ((fn s => [s]), key ^ "-", Importance.Low); 
            val importance = case overrideImportance of 
                                    NONE => defaultImportance 
                                  | SOME i => i; 
            fun makeProp v = QProperty.fromPair 
                                 (Property.fromString (keypre ^ v), 
                                  importance); 
        in 
            qset' (map makeProp (valparser args)) 
        end; 
in 
    filedict' (loadQorRSPropertiesFromFile sets parseRow genProps filename) 
end; 

fun loadRepresentationTable filename = let 
    val sets = (S.empty, S.union); 
    fun parseRow [x, y] = (x, y) 
      | parseRow _ = raise TableError "Malformed representation property entry"; 
    fun genProps (key, args) = 
        let 
            fun findRSGenerator k = SOME (GenDict.get (!rPropertyKeyMap) k) 
                                    handle GenDict.KeyError => NONE; 
            val (valparser, keypre) = case (findRSGenerator key) of 
                                          SOME kt => kt 
                                        | NONE => ((fn s => [s]), key ^ "-"); 
            fun makeProp v = Property.fromString (keypre ^ v); 
        in 
            set' (map makeProp (valparser args)) 
        end; 
in 
    filedict' (loadQorRSPropertiesFromFile sets parseRow genProps filename) 
end; 

end; 
import "util.logging"; 
import "util.set"; 
import "util.dictionary"; 
import "util.csv"; 

import "strategies.property"; 
import "strategies.property_tables"; 
import "strategies.property_readers"; (* Must come after strategies.property_tables *) 
import "strategies.property_importance"; 
import "strategies.property_correspondence"; 

structure RepresentationSelection = 
struct 

structure FileDict = PropertyTables.FileDict; 

(* Read in some data *) 

val propertyTableRep' = ref (FileDict.empty ()); 
val correspondingTable' = ref []; 
val propertyTableQ' = ref (FileDict.empty ()); 

fun init (repTables, corrTables, qTables) = let 
    val _ = Logging.write "\n-- Load the representation tables\n"; 
    val propertyTableRep = 
        foldr (fn (a, b) => FileDict.union a b) 
              (FileDict.empty ()) 
              (map (fn t => (Logging.write ("LOAD " ^ t ^ "\n"); 
                             PropertyTables.loadRepresentationTable t)) repTables) 
        handle FileDict.KeyError => (Logging.error "An RS table has been duplicated"; raise 
FileDict.KeyError); 
    val _ = Logging.write "\n-- Load the correspondence tables\n"; 
    val correspondingTable = 
        foldr (fn (a, b) => a @ b) 
              [] 
              (map (fn t => (Logging.write ("LOAD " ^ t ^ "\n"); 
                             PropertyTables.loadCorrespondenceTable t)) corrTables); 
    val _ = Logging.write "\n-- Load the question tables\n"; 
    val propertyTableQ = 
        foldr (fn (a, b) => FileDict.union a b) 
              (FileDict.empty ()) 
              (map (fn t => (Logging.write ("LOAD " ^ t ^ "\n"); 
                             PropertyTables.loadQuestionTable t)) qTables); 
    fun dedupCorrespondences [] = [] 
      | dedupCorrespondences (x::xs) = let 
          fun removeCorr y [] = [] 
            | removeCorr y (z::zs) = 
              if Correspondence.matchingProperties y z 
              then ( 
                  if Correspondence.equal y z then 
                      zs 
                  else 
                      (Logging.error ("ERROR: Conflicting correspondences:\n"); 
                       Logging.error ("\t" ^ 
                                      (Correspondence.toString y) ^ 
                                      "\n"); 
                       Logging.error ("\t" ^ 
                                      (Correspondence.toString z) ^ 
                                      "\n"); 
                       raise Fail "Conflicting correspondence values") 
              ) 
              else z::(removeCorr y zs); 
      in 
          x::(removeCorr x xs) 
      end; 
in 
    propertyTableRep' := propertyTableRep; 
    correspondingTable' := dedupCorrespondences correspondingTable; 
    propertyTableQ' := propertyTableQ 
end; 

fun propertiesRS rep = 
    FileDict.get (!propertyTableRep') rep 
    handle FileDict.KeyError => 
           (Logging.error ("ERROR: representation '" ^ rep ^ "' not found!\n"); 
           raise FileDict.KeyError); 

fun withoutImportance props = PropertySet.fromList (QPropertySet.map 
(QProperty.withoutImportance) props); 

fun propertiesQ q = 
    FileDict.get (!propertyTableQ') q 
    handle FileDict.KeyError => 
           (Logging.error ("ERROR: question named '" ^ q ^ "' not found!\n"); 
           raise FileDict.KeyError); 

(* 
propInfluence : (question * representation * float) -> (question * representation * float) 
For the given question and representation, adjust the score based on 
their properties. 
*) 
fun propInfluence (q, r, s) = 
    let 
        val _ = Logging.write ("\n"); 
        val _ = Logging.write ("BEGIN propInfluence\n"); 
        val _ = Logging.indent (); 
        val _ = Logging.write ("ARG q = " ^ q ^ " \n"); 
        val _ = Logging.write ("ARG r = " ^ r ^ " \n"); 
        val _ = Logging.write ("ARG s = " ^ (Real.toString s) ^ " \n\n"); 
        val qProps' = propertiesQ q; 
        val rProps = propertiesRS r; 
        val _ = Logging.write ("VAL qProps = " ^ (QPropertySet.toString qProps') ^ "\n"); 
        val _ = Logging.write ("VAL rProps = " ^ (PropertySet.toString rProps) ^ "\n\n"); 
        fun liftImportance c = 
            let 
                fun collateImportances propPairs = 
                    let 
                        val uniqueProperties = PropertySet.fromList (map (fn (p, i) => p) 
propPairs); 
                        fun collectImportances p' [] ans = ans 
                          | collectImportances p' ((p,i)::ps) ans = 
                            if (Property.compare (p', p) = EQUAL) then collectImportances p' 
ps (i::ans) 
                            else collectImportances p' ps ans; 
                    in 
                        PropertySet.map (fn p => (p, collectImportances p propPairs [])) 
uniqueProperties 
                    end; 
                val importanceLookup = (PropertyDictionary.fromPairList o 
                                        collateImportances o 
                                        (map QProperty.toPair) o 
                                        QPropertySet.toList) qProps'; 
                val importanceMax = max Importance.compare; 
                val getImportance = PropertyDictionary.get importanceLookup; 
                val ((qp, _), _, _) = c; 
                val flatten = flatmap (fn x => x); 
                val i = importanceMax (flatten (PropertySet.map getImportance qp)); 
            in 
                (c, i) 
            end; 
        fun modulate strength importance = 
            case importance of 
                Importance.Noise => 0.0 
              | Importance.Zero => 0.0 
              | Importance.Low => 0.33 * strength 
              | Importance.Medium => 0.67 * strength 
              | Importance.High => strength; 
        val qProps = withoutImportance qProps'; 
        val propertyPairs' = List.filter 
                                 (Correspondence.match qProps rProps) 
                                 (!correspondingTable'); 
        val identityPairs = PropertySet.map 
                                (fn p => ((PropertySet.fromList [p], PropertySet.empty ()), 
                                          (PropertySet.fromList [p], PropertySet.empty ()), 
                                          1.0)) 
                              (*)  (PropertySet.intersection qProps rProps);*) 
                                (Correspondence.matchingIntersectionLeft qProps rProps); 
        val identityPairs' = List.filter (fn corr => 
                                             not(List.exists 
                                                     (*(Correspondence.sameProperties corr)*) 
                                                     (Correspondence.matchingProperties corr) 

